
ActiveX Controls

Creating an ActiveX Control is like creating your own custom object or widget.  For example, if you find you want an enhanced command button you can create your own button that can be used in any of your programs.  Or if you are going to use a part of your code many times you may want to think about packaging it into an ActiveX Control.

ActiveX Controls are great ways to reuse code for many different programs.  Each control should have a well-defined API (Advanced Programming Interface) which includes creating and specifying things such as properties and methods.

The properties and methods of your control are what people will use to gain access to it.

Creating ActiveX Controls

The following are steps to create an ActiveX Control:

Start a new ActiveX Control as your New Project in VB.

You should see a window appear for your new ActiveX control.  Everything you place in this window will be your control.

Give your control a name in the properties window like below:

Go to the Project menu and Project Properties.  In the window set a name for your project and also a description of your control.

Now place some standard controls on to your ActiveX Control.  For this example I will draw on a command button.

To add properties and methods you can use the Control Wizard.  To get there click on the Add toolbar button and then User Control.

In the window that appears, select the Wizard and click Open.

The following describes how to add properties, methods, and events with code.

Adding Events

Event declarations are a listing of all events that your control will respond to.  You can add events to the top of your code like this:

'Event Declarations:

Event DblClick()

Event MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single) 

Event MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

Event MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

The events will now be accessible from your main program code just like other controls have events.

You can also call these events in your control’s code.  For instance if someone clicks on a command button, but you want this click to be sent to a control’s click, add this code to your control:

Private Sub Command1_Click()

    Call UserControl_Click

End Sub

‘ if a user clicks on Command1 in your control, it will raise your control’s click event

Adding Properties

To add a property you must first have a variable for it at the top of your code:

'Property Variables:

Dim m_Caption As String

You should also create a constant to hold the default value of your property:

Const m_def_Caption = ""

All properties need Get and Let methods.  Let is automatically called if a property changes in your main code.  Get is automatically called to retrieve a property for your main code.

Public Property Get Caption() As String

    Caption = m_Caption

End Property

Public Property Let Caption(ByVal New_Caption As String)

    m_Caption = New_Caption

    PropertyChanged "Caption"

End Property

Additional code can be added to these two methods if you want other things to happen when a property is changed for instance.

Read and Write properties are sub routines that get called when properties change or are loaded.  It is something VB uses to keep track of given properties.  For each property you create you must have a line in these two subroutines for it.

'Load property values from storage

Private Sub UserControl_ReadProperties(PropBag As PropertyBag)


m_Caption = PropBag.ReadProperty("Caption", m_def_Caption)

End Sub

'Write property values to storage

Private Sub UserControl_WriteProperties(PropBag As PropertyBag)


Call PropBag.WriteProperty("Caption", m_Caption, m_def_Caption)

End Sub

Adding Methods

To add methods to your control, just create a subroutine.  From your main code you can execute this subroutine by using the dot operator.

Public Sub displayMessage(ByVal Message As String)

     Msgbox Message

End Sub

Main code:

MyControl.displayMessage(“Hello”)

Compiling Your Control

To use your control in other programs, go to the file menu and choose Make YourControl.ocx.  Your control will be compiled into an OCX file.

In a standard program you can now add your control to it by adding Components.  You may remember this is done by right clicking on the widget toolbar.  

You will have to find your control in the last of available controls.  Once you have, check it off and it should appear in your toolbar.

Structures

To store records in your program much like a phone book stores information about people you can create structures in Visual Basic.  Each structure could contain information like a name, phone number, and address.  If you wanted to store information about a few people you can create an array of structures.

In VB, a structure can be created with the type operator.

Public Type person

  name As String

  address As String

  phoneNumber As String

End Type

Then I can create an array of my structures:

Dim myAddressBook(20) as person

‘ this would create 20 records to store information for a phone book

To access my records or elements in them I can use the dot operator,

Output = myAddressBook(3).name

‘ this would put person 3’s name into my variable

myAddressBook(3).name = “Joe”

‘ this would set person 3’s name in my variable

Loading Controls

If you would like to create controls or objects while your program is running, the load and unload commands will become very useful.  You can load buttons, menu items, shapes, or even Internet Transfer Controls to allow more than one chat connection.

For example, if I create a shape on a form and name it "shape1". Then I change the index property to 0 for it and its visibility property to false. The first shape will always remain invisible and just exist to be able to create my array of shapes.

Now I have one hidden shape, but it is the start of an array of shapes.

Then in my code I can create shapes and add them to this array of shapes.

For instance, suppose I have some button and whenever somebody clicks this button a new shape is created.

I first need a variable to hold the told number of shapes I have in this array of shapes.

So at the top of my code I add the line:

dim numberOfShapes as Integer

Then in Form Load I add the line:

numberOfShapes = 0

Then I add the following code to Command1_Click:

numberOfShapes = numberOfShapes + 1 

load shape1(numberOfShapes) 

shape1(numberOfShapes).visible = true 

' now my array of shapes would have 2 shapes in it, and the new one would be ‘ visible 

If people keep clicking on the button, new shapes will be added, but numberOfShapes will always know how many shapes I have and keep adding them to the end of my array of shapes. 

Then because I have created objects and allocated memory while my program is running, in order to not have memory leaks, I must remove any objects I created while the program is running. 

So in Form Unload I would stick the following code to remove the objects I created. 

for i=1 to numberOfShapes 

unload shape1(numberofshapes) 

next i 

File Handling

Using files for input and output is a great way for storing data after your program has finished running.  If there is information you would like to save until the next time your program runs you can output it to a file.  Many other file handling functions like copying or deleting files can be performed in Visual Basic as well.  

The following contains sample code for doing such operations.

Checking for File Existence

If Dir$(“c:\myfile.txt”) = “” then     ‘ use the Dir command to search 


Msgbox “File not found”

End if

Listing Files and Folders

‘ This code adds all filenames in a particular directory to a list box

Dim sNextFile as String

SNextFile = Dir$(“c:\data\*.bmp”)

While sNextFile <> “”


PictureList.addItem sNextFile


SNextFile = Dir$

Wend

Copying Files

‘ copying on a local hard drive

FileCopy “D:\My Documents\MyFile.txt”, “C:\Data\MyFile.txt”

‘ copying from a network drive using variables

Dim sDest as String

Dim sSource as String

sSource = “\\myserver\data\myfile.txt”

sdest = “c:\data\myfile.txt”

FileCopy sSource, sDest

Deleting Files

Kill “d:\data\*.doc”

‘ this will delete all doc files in the given directory

Kill “d:\myfile.txt”

‘ this will delete a single file in the given directory

Renaming Files

‘ use the name command to rename files

Name myoldfile.txt As newfilename.txt

‘ move files with the name command and make directories

MkDir “d:\newdirectory”

Name “c:\Windows\myfile.txt” As “d:\data\myfile.txt”

Setting the Current Directory

‘ change the current directory

Chdir “c:\Windows”

‘ change the current drive letter

Chdrive “C:”

Launching Other Programs

‘ launch a file in Notepad

Dim idNumber as double

IdNumber = Shell (“Notepad d:\myfile.txt”, vbNormalFocus)

Sequential File Reading (Input)

To read from a file you first have to open the file.  Once open, you can read line by line.  The example below puts each line from the file into a text box.  Remember to always close your file when done.

It is a good idea to check that the file actually exists before you open it.

Dim sTemp as String 

‘ create a string variable to hold a line from our file

Open “c:\data\myfile.txt” For Input as #1

‘ open a file for input, give it a unique file number

While Not EOF (1)   ‘ while we haven’t reached the end of file #1


Line Input #1, sTemp
‘ put a line from the file into sTemp


ListBox.addItem sTemp   ‘ put the line into a listbox

Wend

Close #1

Sequential File Writing (Output)

To output to a file you first have to open a file to write to.  You have a choice of output mode or append mode.  Appending lets you add text to the end of a file.  Once open, you can write line by line to the file.  Remember to always close your file when done.  

Output Mode

Output Mode creates a file and allows you to output line by line to it.  If you open a file that already exists, the contents of it gets deleted, otherwise your output goes to a new file.

Dim fileNumber as Integer

‘ create a variable to hold our file number

fileNumber = FreeFile

‘ this will actually give you a free number

Open “myfile.txt” for Output as #fileNumber

‘ open a file to write to

Print #fileNumber, “Hello this is a line”

‘ output a line to the file

Close #fileNumber

‘ close the file

Append Mode

Append Mode opens the file specified and allows you to output to the end of it.  Whatever you output to this file gets placed at the end of it.

Dim fileNumber as Integer

‘ create a variable to hold our file number

fileNumber = FreeFile

‘ this will actually give you a free number

Open “myfile.txt” for Append as #fileNumber

‘ open a file to write to

Print #fileNumber, “Hello this is a line”

‘ output a line to the file

Close #fileNumber

‘ close the file

Write vs. Print

The Print command places a new line marker after it prints out each line.  Instead of using the Print command, you can alternatively use the Write command.  The write command will add delimiters and separators.  Other than that, the two are identical.

For example in the file,

Write #fileNumber, “Hello this “,“is a line”
Produces ->  Hello this is a line

Print #fileNumber, “Hello this “,“is a line”
Produces ->  “Hello this “,“is a line”

Databases with ADO

A database can be very useful to store large amounts of information or records, much like a phone book or address book.  By using a Visual Basic control called ADO (ActiveX Data Objects), accessing a database becomes simple.

The following will describe how to setup a connection to an Access database.  A database table must first be setup in Microsoft Access.

Start up Microsoft Access and create a blank database and remember the filename that you saved it as.

You will see a window like this once it has been created:

Click the New button to create a database table.

In the next window, define your fields for your table.  For a phone book, fields would be items such as Name, Phone Number, and Address.

Once you have finished, close the window and save your changes.

Now it is time to hook up your database table to Visual Basic.

Creating a Connection

Your Visual Basic project needs to have the ADO component referenced in order for you to be able to use ADO.  To add the component go to the Project menu and then References.

Then you must check off the control for ADO in the following list.  It is called Microsoft ActiveX Data Objects 2.5 Library.

The ADO Control comes with the Visual Studio SP4.  If the ADO Control is not listed for you, then you need to download the Visual Studio Service Pack 4 from Microsoft’s web site (it is a free download).

To create a connection to your database table, add the following line of code above all of your subroutines or in a module:

Public mydb as new adodb.connection

myDB is the name that we will use to refer to our database object from now on.

In the form load of your first form, add the following line of code to actually open up the database:

myDB.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & _

App.Path & "\yourdatabase.mdb;Persist Security Info=False"

yourdatabase.mdb should be replaced by the filename that you saved your database under in Microsoft Access.  The line of code also assumes that your database file is saved in the same directory as your VB project files (the application path).

In the form unload section of your program, you need to remember to close the database you opened:

myDB.close

Recordsets

A recordset is simply a group of records.  Creating a recordset is helpful to select a certain group of records from your database.  These records could have certain fields in common.  For example, you could select all people from a phone book with the name John.  This would create a recordset.

Record selection is usually done using SQL which is a database query language.

To select records and create a record set, first declare a recordset in your code like this:

Dim myRS As New adodb.Recordset

Then you can select records to be in this recordset by using SQL and the open command:

myRS.Open "Select * from PhoneBook", myDB

This selection statement will add records to your recordset from the table PhoneBook in the database myDB.  In this example, all the records from the table PhoneBook will be added to the recordset.

The * tells Visual Basic to grab all the fields of the chosen records.

Because this isn’t a course in SQL, only a few more common selection statements will be given.  To search for records with a given field the SQL looks like the following.

myRS.Open “Select * from PhoneBook where FirstName=’John’”, myDB

This would select all records from the table PhoneBook where the field FirstName was John.

You can combine more than one field like this:

myRS.Open “Select * from PhoneBook where FirstName=’John’ and Age=20”, myDB

You could also use text from a textbox to replace the name John:

myRS.Open “Select * from PhoneBook where FirstName=’ & txtName.text & “’”, myDB

Once you have a recordset with records, you can step through each record and perform some operation.  A common one may be to add each record to a list box.

The MoveNext command will move to the next record in your recordset and keep doing this until the end of the recordset is reached, indicated by myRS.EOF
While myRS.EOF = False

myRS.MoveNext

Wend

Additional recordset commands include:

.MoveFirst

.MoveLast

.MovePrev
After I am done performing operations on my recordset, I must close it:
myRS.close

Adding Records

Use the following code to add records to your database:

Dim myRS As New adodb.Recordset

myRS.Open "PhoneBook", myDB, adOpenKeyset, adLockOptimistic

‘ PhoneBook is the name of my table

‘ myDB is the name of my database

myRS.AddNew

‘ create a new entry in the database

myRS!Name = “John”

myRS!Address = “123 Main St.”

‘ fill in the new entries two fields

myRS.Update

myRS.Requery

myRS.Close

‘ finish the update and close the recordset

Editing Records

To edit records you must first search for the record you want to update.  Once found you can change any fields you wish.  Use the following code to update or edit records in your database:

Dim newRS As New adodb.Recordset

newRS.Open "PhoneBook", myDB, adOpenKeyset, adLockOptimistic

‘ PhoneBook is the name of my table

‘ myDB is the name of my database

newRS.Find "Name=’John’”

‘ search for the record to update based on the first name

newRS.Fields("Name").Value = “Sue”

‘ change the name field of John’s record to now be Sue

newRS.Update

newRS.Close

‘ finish the update and close the database

Deleting Records

To delete one or more records you will be introduced to another SQL command.

Dim mySQL as String

‘ create a string variable

mySQL = "Delete * from PhoneBook where Name=’John’”

‘ this SQL command will delete all records from the table PhoneBook where the ‘ field Name is John

myDB.Execute (mySQL)

‘ execute the SQL command we put in our variable mySQL
Creating Custom Web Browsers

Visual Basic 6 comes with a Web Browser control that makes it easy to create a custom web browser.  

To add a web browser click on the add form button.

Then click on the Web Browser icon.

The web browser form comes with a combo box to enter in URLs, a picture box to view a web site in, a toolbar with six buttons, an image list that holds the pictures for the six buttons, and a timer control which is used to display status messages.

The properties of the buttons in the toolbar can be further customized by setting their enable property.

In the form load event procedure add the following code:

tbToolBar.Buttons(“Back”).Enabled = False

tbToolBar.Buttons(“Forward”).Enabled = False

tbToolBar.Buttons(“Stop”).Enabled = False

tbToolBar.Buttons(“Refresh”).Enabled = False

In the brwWebBrowser_NavigateComplete2 even procedure, add the following code:

tbToolBar.Buttons(“Refresh”).Enabled = True

Type the following code at the end of the code window to enable and disable the forward and back buttons when appropriate:

Private Sub brwWebBrowser_CommandStateChange _


(ByVal Command As Long, ByVal Enable As Boolean)

If Command = CSC_NAVIGATEBACK Then


tbToolBar.Buttons(“Back”).Enabled = Enable

End If

If Command = CSC_NAVIGATEFORWARD Then


tbToolBar.Buttons(“Forward”).Enabled = Enable

End If

End Sub

Also, add the following code to the timer’s event procedure to enable and disable the stop button when appropriate:

Private Sub timTimer_Timer()

If brwWebBrowser.Busy = False Then


timTimer.Enabled = False


Me.Caption = brwWebBrowser.LocationName


tbToolBar.Buttons(“Stop”).Enabled = False

Else


Me.Caption = “Working…”


tbToolBar.Buttons(“Stop”).Enabled = True

End If

End Sub

Internet Transfer Control

The Internet Transfer Control (ITC) is used to allow Internet data transfer in your Visual Basic programs.  For example, ITC, can be used in FTP or to retrieve the text of a web page without displaying it.

To add ITC to your program, right click on the main toolbar and click on Add Components in the menu that pops up.

Select Microsoft Internet Transfer Control 6.0 from the list of controls. An icon will appear in your toolbar for ITC.  Double click on it to add ITC to your form.

Here is some sample code to show you how ITC works:

eg.

‘ following is code that will receive the HTML text from a web page

If txtURL.Text <> “” Then


RichTextBox1.Text = Inet1.OpenURL(txtURL.Text, icString)

End If

‘ a URL is contained in txtURL and the HTML text will be displayed in 

‘ a rich text box named RichTextBox1

OpenURL Method

The OpenURL method opens a URL for manipulation.

It’s syntax is:

Inet1.OpenURL(URL, DataType)
URL is the full URL to open, it must either be an HTTP server or an FTP server.

DataType is the type of data to be retrieved. 


It can be either text (icString) or binary (icByteArray).

The return value of OpenURL is the data returned by the URL.  Data transfer with OpenURL is synchronous, your program will pause until the data transfer is complete.

eg.

‘ retrieving a binary file and saving it to disk

Dim b() as Byte

Dim strURL As String, fn As Integer

strURL = “ftp://ftp.someserver.com/somefile.zip”
‘ Retrieve the file as a byte array

b() = Inet1.OpenURL(strURL, icByteArray)

fn = Freefile

Open “C:\somefile.zip” For Binary Access Write As #fn

Put #fn, , b()

Close #fn

StateChanged Event

The only event ITC has is StateChanged.  The syntax for the StateChanged event procedure is:

Inet1_StateChanged(ByVal NewState As Integer)

NewState is a value showing what state the ITC is in.  Possible states include:

	Constant
	Value
	Status

	IcNone

IcResolvingHost

IcHostResolved

IcConnecting

icConnected

IcRequesting

IcRequestSent

IcReceivingResponse

IcResponseReceived

IcDisconnecting

IcDisconnected

IcError

IcResponseCompleted
	0

1

2

3

4

5

6

7

8

9

10

11

12
	None to report

Looking up IP address of host

Fount IP address of host

Connecting to host

Connected to host

Sending request to host

Successfully sent request

Receiving response from host

Received response from host

Disconnecting from host

Successfully disconnected

An error occurred

Request complete, data received


The StateChanged event can be used for supplying status messages for the user.

GetChunk Method

To receive data asynchronously (your program can continue working while data is being received) you can use the GetChunk Method.

The syntax is:

Inet1.GetChunk(datasize,[datatype])

Datasize is of type long that specifies  the number of bytes to retrieve at a time.  Retrieve a small amount of bytes at a time for efficiency (eg. 512 or 1024 bytes).

Datatype is optional and specifies whether the data is binary or text.  The constants icString and icByte are used again.

eg.

Prive Sub Inet1_StateChanged(ByVal State As Integer)

Dim temp1, temp2

Select Case State


‘ Other State cases not shown.


Case icResponseCompleted



temp1 = “”



temp2 = “”



Do





temp1 = Inet.GetChunk(412, icString)




temp2 = temp2 & temp1



Loop Until temp1 = “”      ‘ loop until buffer is empty



RichTextBox1.Text = temp2

End Select

End Sub

Execute Method

The execute method can be used to send commands to the remote computer.  The syntax is:

Inet1.Execute (URL, command, data, requestHeaders)

URL specifies the computer to which the command is being sent

command is the command to be sent


GET retrieves data from the server



(the most commanly used method, stores data in the ITC’s buffer, use previous



examples to retrieve this information from the buffer)


HEAD retrieves header information from the server


POST sends information required to complete a request


PUT uploads a file to the server

data and requestHeaders are for more difficult commands being sent to the server


eg.


Inet1.Execute “http://www.someplace.com/myfile.zip”, “GET”

Properties

The ITC has specific properties that may be useful:

· AccessType
· icUseDefault (value = 0) uses settings from the Windows registry

· icDirect (value = 1) direct connection to the Internet

· icNamedProxy(value = 2) uses a proxy server

· Password
· specifies the password for logging on to a server

· Protocol
· IcUnknown (value = 0) unknown protocol

· IcDefault (value = 1) default protocol

· IcFTP (value = 2) file transfer protocol

· IcReserved (value = 3) reserved for future use

· IcHTTP (value = 4) Hypertext Transfer Protocol

· IcHTTPS (value = 5) Secure Hypertext Transfer Protocol

· RequestTimeout
· time, in seconds, before the server time-outs while waiting for a response

· UserName

· specifies a username to logon to a server, anonymous Is sent if this property is left blank

Chat

Creating a chat program in Visual Basic is actually fairly simple.  It uses the Internet Transfer Control to send and receive information.  Here are some steps and code to setup a simple chat program taken from the MSDN help on the Winsock Control:

A WinSock control allows you to connect to a remote machine and exchange data using either the User Datagram Protocol (UDP) or the Transmission Control Protocol (TCP). Both protocols can be used to create client and server applications. Like the Timer control, the WinSock control doesn't have a visible interface at run time.

When using the WinSock control, the first consideration is whether to use the TCP or the UDP protocol. TCP requires a connection to be made between the sender and receiver.  UDP does not require a connection to first be made.  TCP is good if you want to be positive that information is sent and is better for sending to a small amount of users.  UDP is better for small amounts of data that may get sent to many users. 

Setting the Protocol

To set the protocol that your application will use: at design-time, on the Properties window, click Protocol and select either sckTCPProtocol, or sckUDPProtocol. You can also set the Protocol property in code, as shown below:

Winsock1.Protocol = sckTCPProtocol

Determining the Name of Your Computer

To connect to a remote computer, you must know either its IP address or its "friendly name." The IP address is a series of three digit numbers separated by periods (xxx.xxx.xxx.xxx). In general, it's much easier to remember the friendly name of a computer. 

To find your computer's name:
On the Taskbar of your computer, click Start.

On the Settings item, click the Control Panel.

Double-click the Network icon.

Click the Identification tab.

The name of your computer will be found in the Computer name box. 

Once you have found your computer's name, it can be used as a value for the RemoteHost property.

TCP Connection Basics

When creating an application that uses the TCP protocol, you must first decide if your application will be a server or a client. Creating a server means that your application will "listen," on a designated port. When the client makes a connection request, the server can then accept the request and thereby complete the connection. Once the connection is complete, the client and server can freely communicate with each other.

The following steps create a rudimentary server:

To Create a TCP Server 

Create a new Standard EXE project.

Change the name of the default form to frmServer.

Change the caption of the form to "TCP Server."

Draw a Winsock control on the form and change its name to tcpServer.

Add two TextBox controls to the form. Name the first txtSendData, and the second txtOutput.

Add the code below to the form. 

Private Sub Form_Load()

    ' Set the LocalPort property to an integer.

    ' Then invoke the Listen method.

    tcpServer.LocalPort = 1001

    tcpServer.Listen 

    frmClient.Show ' Show the client form.

End Sub

Private Sub tcpServer_ConnectionRequest _

(ByVal requestID As Long)

    ' Check if the control's State is closed. If not, 

    ' close the connection before accepting the new

    ' connection.

    If tcpServer.State <> sckClosed Then _

    tcpServer.Close

    ' Accept the request with the requestID 

    ' parameter.

    tcpServer.Accept requestID

End Sub

Private Sub txtSendData_Change()

    ' The TextBox control named txtSendData 

    ' contains the data to be sent. Whenever the user 

    ' types into the  textbox, the  string is sent 

    ' using the SendData method.

    tcpServer.SendData txtSendData.Text

End Sub

Private Sub tcpServer_DataArrival _

(ByVal bytesTotal As Long)

    ' Declare a variable for the incoming data. 

    ' Invoke the GetData method and set the Text

    ' property of a TextBox named txtOutput to 

    ' the data.

    Dim strData As String

    tcpServer.GetData strData

    txtOutput.Text = strData

End Sub

The procedures above create a simple server application. However, to complete the scenario, you must also create a client application.

To Create a TCP Client 

Add a new form to the project, and name it frmClient.

Change the caption of the form to TCP Client.

Add a Winsock control to the form and name it tcpClient.

Add two TextBox controls to frmClient. Name the first txtSend, and the second txtOutput.

Draw a CommandButton control on the form and name it cmdConnect.

Change the caption of the CommandButton control to Connect.

Add the code below to the form. 
Important   Be sure to change the value of the RemoteHost property to the friendly name of your computer.

Private Sub Form_Load()

    ' The name of the Winsock control is tcpClient.

    ' Note: to specify a remote host, you can use 

    ' either the IP address (ex: "121.111.1.1") or

    ' the computer's "friendly" name, as shown here.

    tcpClient.RemoteHost = "RemoteComputerName"

    tcpClient.RemotePort = 1001

End Sub

Private Sub cmdConnect_Click()

    ' Invoke the Connect method to initiate a 

    ' connection.

    tcpClient.Connect

End Sub

Private Sub txtSendData_Change()

    tcpClient.SendData txtSend.Text

End Sub

Private Sub tcpClient_DataArrival _

(ByVal bytesTotal As Long)

    Dim strData As String

    tcpClient.GetData strData

    txtOutput.Text = strData

End Sub

The code above creates a simple client-server application. To try the two together, run the project, and click Connect. Then type text into the txtSendData TextBox on either form, and the same text will appear in the txtOutput TextBox on the other form.

Accepting More than One Connection Request

The basic server outlined above accepts only one connection request. However, it is possible to accept several connection requests using the same control by creating a control array. In that case, you do not need to close the connection, but simply create a new instance of the control (by setting its Index property), and invoking the Accept method on the new instance.

The code below assumes there is a Winsock control on a form named sckServer, and that its Index property has been set to 0; thus the control is part of a control array. In the Declarations section, a module-level variable intMax is declared. In the form's Load event, intMax is set to 0, and the LocalPort property for the first control in the array  is set to 1001. Then the Listen method is invoked on the control, making it the "listening control. As each connection request arrives, the code tests to see if the Index is 0 (the value of the "listening" control). If so, the listening control increments intMax, and uses that number to create a new control instance. The new control instance is then used to accept the connection request.

Private intMax As Long

Private Sub Form_Load()

   intMax = 0

   sckServer(0).LocalPort = 1001

   sckServer(0).Listen

End Sub

Private Sub sckServer_ConnectionRequest _

(Index As Integer, ByVal requestID As Long)

   If Index = 0 Then

      intMax = intMax + 1

      Load sckServer(intMax)

      sckServer(intMax).LocalPort = 0

      sckServer(intMax).Accept requestID

      Load txtData(intMax)

   End If

End Sub

UDP Basics

Creating a UDP application is even simpler than creating a TCP application because the UDP protocol doesn't require an explicit connection. In the TCP application above, one Winsock control must explicitly be set to "listen," while the other must initiate a connection with the Connect method.

In contrast, the UDP protocol doesn't require an explicit connection. To send data between two controls, three steps must be completed (on both sides of the connection): 

Set the RemoteHost property to the name of the other computer.

Set the RemotePort property to the LocalPort property of the second control.

Invoke the Bind method specifying the LocalPort to be used. (This method is discussed in greater detail below.) 
Because both computers can be considered "equal" in the relationship, it could be called a peer-to-peer application. To demonstrate this, the code below creates a "chat" application that allows two people to "talk" in real time to each other:

To Create a UDP Peer 

Create a new Standard EXE project.

Change the name of the default form to frmPeerA.

Change the caption of the form to "Peer A."

Draw a Winsock control on the form and name it udpPeerA.

On the Properties page, click Protocol and change the protocol to UDPProtocol.

Add two TextBox controls to the form. Name the first txtSend, and the second txtOutput.

Add the code below to the form. 
Private Sub Form_Load()

    ' The control's name is udpPeerA

    With udpPeerA

        ' IMPORTANT: be sure to change the RemoteHost 

        ' value to the name of your computer.

        .RemoteHost= "PeerB" 

        .RemotePort = 1001   ' Port to connect to.

        .Bind 1002                ' Bind to the local port.

    End With

    frmPeerB.Show                 ' Show the second form.

End Sub

Private Sub txtSend_Change()

    ' Send text as soon as it's typed.

    udpPeerA.SendData txtSend.Text

End Sub

Private Sub udpPeerA_DataArrival _

(ByVal bytesTotal As Long)

    Dim strData As String

    udpPeerA.GetData strData

    txtOutput.Text = strData

End Sub

To Create a Second UDP Peer 

Add a standard form to the project.

Change the name of the form to frmPeerB.

Change the caption of the form to "Peer B."

Draw a Winsock control on the form and name it udpPeerB.

On the Properties page, click Protocol and change the protocol to UDPProtocol.

Add two TextBox controls to the form. Name the TextBox txtSend, and the second txtOutput.

Add the code below to the form. 
Private Sub Form_Load()

    ' The control's name is udpPeerB.

    With udpPeerB

        ' IMPORTANT: be sure to change the RemoteHost 

        ' value to the name of your computer.

        .RemoteHost= "PeerA"

        .RemotePort = 1002    ' Port to connect to.

        .Bind 1001                ' Bind to the local port.

    End With

End Sub

Private Sub txtSend_Change()

    ' Send text as soon as it's typed.

    udpPeerB.SendData txtSend.Text

End Sub

Private Sub udpPeerB_DataArrival _

(ByVal bytesTotal As Long)

    Dim strData As String

    udpPeerB.GetData strData

    txtOutput.Text = strData

End Sub
To try the example, press F5 to run the project, and type into the txtSend TextBox on either form. The text you type will appear in the txtOutput TextBox on the other form.

About the Bind Method

As shown in the code above, you must invoke the Bind method when creating a UDP application. The Bind method "reserves" a local port for use by the control. For example, when you bind the control to port number 1001, no other application can use that port to "listen" on. This may come in useful if you wish to prevent another application from using that port.

The Bind method also features an optional second argument. If there is more than one network adapter present on the machine, the LocalIP argument allows you to specify which adapter to use. If you omit the argument, the control uses the first network adapter listed in the Network control panel dialog box of the computer's Control Panel Settings.

When using the UDP protocol, you can freely switch the RemoteHost and RemotePort properties while remaining bound to the same LocalPort. However, with the TCP protocol, you must close the connection before changing the RemoteHost and RemotePort properties.

